Homework set 7

Jacopo Ceccuti s215158

Problem 1

```
> restart:with(LinearAlgebra):with(plots):
```

A)

The parametrization of the solid body is given by:

```
> Omega:=v*cos(u),v*sin(u),w
```

$$\Omega := v\cos(u),\, v\sin(u),\, w \tag{1}$$

With the intervals: $u \in \left[0, \frac{\pi}{2}\right]$, $v \in [0, a]$ and $w \in [0, b]$

B)

We are given a vector field **V** and asked to compute the flux out through the surface $\partial\Omega$ of Ω:

```
> V:=unapply(<x^2-2*x*z,5*y*z,2*y*exp(x)>,x,y,z):
  'V(x,y,z)'=V(x,y,z)
```

$$V(x, y, z) = \begin{bmatrix} x^2 - 2\,x\,z \\ 5\,y\,z \\ 2\,y\,e^x \end{bmatrix} \tag{2}$$

We can compute the Flux in two ways:

- Using Gauss's theorem
- Summing up the fluxes through the different surfaces

Let's use Gauss's theorem first:

$Flux = \int_{\Omega} Div(V)\, d\mu = \int_{\delta\Omega} V \cdot n\, d\mu$ The right hand side in particular

Before we can compute the triple integral we have to find the divergence of the vector field:

```
> Div:=unapply(diff(V(x,y,z)[1],x)+diff(V(x,y,z)[2],y)+diff(V(x,y,
  z)[3],z),x,y,z)
```

$$Div := (x, y, z) \mapsto 2 \cdot x + 3 \cdot z \tag{3}$$

And the Jacobian function for our parametrization:

```
> Jac:=abs(simplify(Determinant(<diff(<Omega>,u)|diff(<Omega>,v)
  |diff(<Omega>,w)>)))
```

$$Jac := |v| \tag{4}$$

Now it's possible to compute the integral:

```
> Flux_G:=int(Div(Omega)*Jac,u=0..Pi/2,v=0..a,w=0..b) assuming a>=
  0, b>=0
```

$$Flux\_G := \frac{2}{3}\, a^3\, b + \frac{3}{8}\, \pi\, a^2\, b^2 \tag{5}$$

Now let's find the flux of the three different surfaces and sum them together at the end:

$Flux = \int_{\delta\Omega} V \cdot n\, d\mu$

Attention: *the normal vector must point outward.*

```
> top:=v*cos(u),v*sin(u),b
```

$$top := v\cos(u),\, v\sin(u),\, b \tag{6}$$

```
> bottom:=v*cos(u),v*sin(u),0
```

$$bottom := v\cos(u),\ v\sin(u),\ 0 \tag{7}$$

```
> shell:=a*cos(u),a*sin(u),w
```

$$shell := a\cos(u),\ a\sin(u),\ w \tag{8}$$

```
> lateral_1:=0,a*u,w
```

$$lateral\_1 := 0,\ a\,u,\ w \tag{9}$$

```
> lateral_2:=a*u,0,w
```

$$lateral\_2 := a\,u,\ 0,\ w \tag{10}$$

```
> Ntop:=simplify(CrossProduct(diff(<top>,v),diff(<top>,u)))
```

$$Ntop := \begin{bmatrix} 0 \\ 0 \\ v \end{bmatrix} \tag{11}$$

```
> Nbottom:=simplify(CrossProduct(diff(<bottom>,u),diff(<bottom>,v))
  )
```

$$Nbottom := \begin{bmatrix} 0 \\ 0 \\ -v \end{bmatrix} \tag{12}$$

```
> Nshell:=simplify(CrossProduct(diff(<shell>,u),diff(<shell>,w)))
```

$$Nshell := \begin{bmatrix} a\cos(u) \\ a\sin(u) \\ 0 \end{bmatrix} \tag{13}$$

```
> Nlateral_1:=simplify(CrossProduct(diff(<lateral_1>,w),diff
  (<lateral_1>,u)))
```

$$Nlateral\_1 := \begin{bmatrix} -a \\ 0 \\ 0 \end{bmatrix} \tag{14}$$

```
> Nlateral_2:=simplify(CrossProduct(diff(<lateral_2>,u),diff
  (<lateral_2>,w)))
```

$$Nlateral\_2 := \begin{bmatrix} 0 \\ -a \\ 0 \end{bmatrix} \tag{15}$$

```
> Flux_top:=int(V(top).Ntop,u=0..Pi/2,v=0..a)
```

$$Flux\_top := 2 + 2\,a\,e^{a} - 2\,e^{a} - a^{2} \tag{16}$$

```
> Flux_bottom:=int(V(bottom).Nbottom,u=0..Pi/2,v=0..a)
```

$$Flux\_bottom := -2 - 2\,a\,e^{a} + 2\,e^{a} + a^{2} \tag{17}$$

```
> Flux_shell:=int(V(shell).Nshell,u=0..Pi/2,w=0..b) assuming a>0,
  b>0
```

$$Flux\_shell := \frac{2}{3}\,a^{3}\,b + \frac{3}{8}\,\pi\,a^{2}\,b^{2} \tag{18}$$

```
> Flux_lateral_1:=int(V(lateral_1).Nlateral_1,u=0..1,w=0..b)
```

$$Flux\_lateral\_1 := 0 \tag{19}$$

```
> Flux_lateral_2:=int(V(lateral_2).Nlateral_2,u=0..1,w=0..b)
```

$$Flux\_lateral\_2 := 0 \tag{20}$$

```
> Flux:=Flux_top+Flux_bottom+Flux_shell+Flux_lateral_1+
  Flux_lateral_2
```

$$Flux := \frac{2}{3}\, a^3\, b + \frac{3}{8}\, \pi\, a^2\, b^2 \tag{21}$$

The two results are matching.
A plot for the situation represented in the problem is:

```
> shell_part:=plot3d(<cos(u),sin(u),w>,u=0..Pi/2,w=0..5,style=
  patchnogrid,color="Nautical GrayViolet"):
> top_part:=plot3d(<v*cos(u),v*sin(u),5>,u=0..Pi/2,v=0..1,style=
  patchnogrid,color="Nautical GrayViolet"):
> bottom_part:=plot3d(<v*cos(u),v*sin(u),0>,u=0..Pi/2,v=0..1,style=
  patchnogrid,color="Nautical GrayViolet"):
> vector_field:=fieldplot3d(V(x,y,z),x=-5..5,y=-5..5,z=0..7,arrows=
  THICK,grid=[10,10,10],color=red):
> LT_1:=plot3d(<0,u,w>,u=0..1,w=0..5,style=patchnogrid,color=
  "Nautical GrayViolet"):
> LT_2:=plot3d(<u,0,w>,u=0..1,w=0..5,style=patchnogrid,color=
  "Nautical GrayViolet"):
> display(shell_part,top_part,bottom_part,vector_field,LT_1,LT_2,
  scaling=constrained,labels=[x,y,z],view=[-2..2,-2..2,0..5])
```



Problem 2

```
> restart:with(LinearAlgebra):with(plots):
```

A)

In (x, y, z) space we consider the filled-in triangle T with the corners in: A(0, 0, 3), B(0, 2, 0) and C(1, 0, 0). To parametrize it we first start with the segment BC on the (x,y) plane and then try to find all the points AP with P such that: B≤P≤C.

```
> s:=(v*(3-u))/3,((3-u)*(2-2*v))/3,u
```

$$s := \frac{v(3-u)}{3}, \frac{(3-u)(2-2v)}{3}, u \tag{22}$$

With the intervals: $u \in [0, 3]$, $v \in [0, 1]$

B)

The formula for the circulation using Stoke's theorem is:

$Circulation = \int_S Curl(V) \cdot N \, d\mu = \int_{\delta S} V \cdot e \, d\mu$ In particular we are gonna first use the left hand side

To find the circulation we need to find the normal vector to the surface and verify that the direction agrees with the right hand rule.

```
> N:=simplify(CrossProduct(diff(<s>,v),diff(<s>,u)))
```

$$N := \begin{bmatrix} -2 + \frac{2u}{3} \\ -1 + \frac{u}{3} \\ -\frac{2}{3} + \frac{2u}{9} \end{bmatrix} \tag{23}$$

The vector field is given by:

```
> V:=<z,x,y>
```

$$V := \begin{bmatrix} z \\ x \\ y \end{bmatrix} \tag{24}$$

The curl of it is found doing:

```
> Curl:=unapply(simplify(<diff(V[3],y)-diff(V[2],z),diff(V[1],z)-
  diff(V[3],x),diff(V[2],x)-diff(V[1],y)>),x,y,z)
```

$$Curl := (x, y, z) \mapsto \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix} \tag{25}$$

Finally is possible to apply the formula and find the circulation

```
> Circ:=int(Curl(V).N,u=0..3,v=0..1)
```

$$Circ := -\frac{11}{2} \tag{26}$$

Now we can try to verify Stoke's theorem calculating the circulation for every side and sum them together

The orientation needs to be: ABCA

```
> restart:with(LinearAlgebra):with(plots):
> V:=unapply(<z,x,y>,x,y,z):
  'V(x,y,z)'=V(x,y,z)
```

$$V(x,y,z) = \begin{bmatrix} z \\ x \\ y \end{bmatrix} \tag{27}$$

From B to C

```
> p:=v,2-2*v,0
```

$$p := v, 2 - 2\,v, 0 \tag{28}$$

With $v \in [0, 1]$

```
> plot3d(<p>,v=0..1,labels=[x,y,z])
```



From C to A

```
> d:=1-u/3,0,u
```

$$d := 1 - \frac{u}{3}, 0, u \tag{29}$$

With $u \in [0, 3]$

```
> plot3d(<d>,u=0..3,labels=[x,y,z])
```



From A to B

```
> j:=0,w,3-3*w/2
```

$$j := 0, w, 3 - \frac{3\,w}{2} \tag{30}$$

With $w \in [0, 2]$

```
> plot3d(<j>,w=0..2,labels=[x,y,z])
```



Now the circulation is computed using the following:

$$Circ = Tan(V, p) + Tan(V, d) + Tan(V, j) = \int_0^1 V(p) \cdot p' \, \mathrm{d}v + \int_0^3 V(d) \cdot d' \, \mathrm{d}u + \int_0^2 V(j) \cdot j' \, \mathrm{d}w$$

```
> int(V(p).diff(<p>,v),v=0..1)+int(V(d).diff(<d>,u),u=0..3)+int(V
  (j).diff(<j>,w),w=0..2)
```

$$-\frac{11}{2} \tag{31}$$

Which is equal to the one found with Stoke's theorem

Problem 3

```
> restart:with(LinearAlgebra):with(plots):
```

A)

Consider in (x, y, z) space the vector field **V**(x, y, z) = (2x + 3y, 2y + 3x, -4z) and the function:

$$F(x, y, z) = a \cdot x^2 + b \cdot y^2 + c \cdot z^2 + d \cdot x \cdot y, \ (x, y, z) \in \mathbb{R}^3$$

```
> F(x,y,z):=a*x^2+b*y*2+c*z^2+d*x*y
```

$$F(x, y, z) := a\, x^2 + c\, z^2 + d\, x\, y + 2\, b\, y \tag{32}$$

```
> V:=unapply(<2*x+3*y,2*y+3*x,-4*z>,x,y,z):
  'V(x,y,z)'=V(x,y,z)
```

$$\tag{33}$$

$$V(x, y, z) = \begin{bmatrix} 2\,x + 3\,y \\ 2\,y + 3\,x \\ -4\,z \end{bmatrix} \tag{33}$$

In order to find the constants we have to solve the set made out of these three equations:

```
> dF/dx=V[1];
  dF/dy=V[2];
  dF/dz=V[3]
```

$$\frac{dF}{dx} = V_1$$
$$\frac{dF}{dy} = V_2$$
$$\frac{dF}{dz} = V_3 \tag{34}$$

To get:

```
> diff(F(x,y,z),x)=2*x+3*y;
  diff(F(x,y,z),y)=2*y+3*x;
  diff(F(x,y,z),z)=-4*z
```

$$2\,a\,x + d\,y = 2\,x + 3\,y$$
$$d\,x + 2\,b = 2\,y + 3\,x$$
$$2\,c\,z = -4\,z \tag{35}$$

That can be easily solved looking at the constats to get:

```
> a:=1;
  b:=1;
  c:=-2;
  d=3
```

$$a := 1$$
$$b := 1$$
$$c := -2$$
$$d = 3 \tag{36}$$

B)

The parametrization of the right half of the unit circle in the (y, z) plane centered at (0, 1, 1) is:

```
> s:=0,v*cos(u)+1,v*sin(u)+1
```

$$s := 0,\, v\cos(u) + 1,\, v\sin(u) + 1 \tag{37}$$

With the intervals: $u \in \left[-\frac{\pi}{2}, \frac{\pi}{2}\right], v \in [0, 1]$

Here it's visualized together with the vector field **V:**

```
> vector_field:=fieldplot3d(V(x,y,z),x=-2..2,y=-2..2,z=-2..2,
  arrows=SLIM,grid=[10,10,10],color=red):
```

```
> surface:=plot3d(<s>,v=0..1,u=-Pi/2..Pi/2,style=patchnogrid,color=
  "Nautical GrayViolet",transparency=0.3):
```

```
> display(vector_field,surface,scaling=constrained,labels=[x,y,z],
  view=[-1/2..1/2,0..2,-1/2..2])
```



From the parametrization for the unit circle the parametrization for the other line of it is easily found:

```
> k:=0,cos(u)+1,sin(u)+1
```

$$k := 0, \cos(u) + 1, \sin(u) + 1 \quad \textbf{(38)}$$

With $u \in \left[-\frac{\pi}{2}, \frac{\pi}{2} \right]$

Here it's visualized, in green, with the unit circle and the vector field:

```
> K:=spacecurve(<k>,u=-Pi/2..Pi/2,color=green,thickness=4,
  numpoints=1000):
> display(vector_field,K,surface,scaling=constrained,labels=[x,y,
  z],view=[-1/2..1/2,0..2,-1/2..2])
```



We need to find the other piece that corresponds to the diameter of the circle:

```
> kk:=0,1,2-u
```

$$kk := 0, 1, 2 - u \qquad (39)$$

With $u \in [0, 2]$

Here it's visualized, in blue, with the unit circle, the vector field and the other boundary curve:

```
> KK:=spacecurve(<kk>,u=0..2,color=blue,thickness=4,numpoints=1000)
  :
> display(vector_field,K,KK,surface,scaling=constrained,labels=[x,
  y,z],view=[-1/2..1/2,0..2,-1/2..2])
```



The formula for the tangential line integral of **V** along k is:

$Tan(V, k) = \int_k V \cdot e \, \mathrm{d}\mu$ Where e is the tangential vector

```
> e_k:=diff(<k>,u)
```

$$e\_k := \begin{bmatrix} 0 \\ -\sin(u) \\ \cos(u) \end{bmatrix} \quad \textbf{(40)}$$

```
> Tan_k:=int(V(k).e_k,u=-Pi/2..Pi/2)
```

$$Tan\_k := -8 \quad \textbf{(41)}$$

```
> e_kk:=diff(<kk>,u)
```

$$e\_kk := \begin{bmatrix} 0 \\ 0 \\ -1 \end{bmatrix} \quad \textbf{(42)}$$

```
> Tan_kk:=int(V(kk).e_kk,u=0..2)
```

$$Tan\_kk := 8 \quad \textbf{(43)}$$

```
> Circ:=Tan_k+Tan_kk
```

$$Circ := 0 \quad \textbf{(44)}$$

Now let's check the result with the Stoke's theorem:

$$Circulation = \int_S Curl(V) \cdot N \, \mathrm{d}\mu = \int_{\delta S} V \cdot e \, \mathrm{d}\mu$$

```
> Curl:=unapply(simplify(<diff(V[3],y)-diff(V[2],z),diff(V[1],z)-diff(V[3],x),diff(V[2],x)-diff(V[1],y)>),x,y,z)
```

$$Curl := (x, y, z) \mapsto \begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix} \tag{45}$$

```
> N:=simplify(CrossProduct(diff(<s>,v),diff(<s>,u)))
```

$$N := \begin{bmatrix} v \\ 0 \\ 0 \end{bmatrix} \tag{46}$$

```
> Circ_S:=int(Curl(V).N,u=-Pi/2..Pi/2,v=0..1)
```

$$Circ\_S := 0 \tag{47}$$

The two results seem to match

C)

Show that the vector field $\mathbf{W}(x, y, z) = \left(2\,yz + xz - x^2, -2\,xz - yz + y^2, y^2 - x^2 + z^2\right)$ is a vector potential for **V**.

In order to have a vector potential **V** must have its divergence equal to zero, let's verify that first:

```
> Div:=unapply(diff(V(x,y,z)[1],x)+diff(V(x,y,z)[2],y)+diff(V(x,y,z)[3],z),x,y,z)
```

$$Div := (x, y, z) \mapsto 0 \tag{48}$$

And this equation needs to hold true:

$\nabla \times \mathbf{W} = \mathbf{V}$

Expanded version:

$$\begin{bmatrix} \frac{\delta}{\delta x} \\ \frac{\delta}{\delta y} \\ \frac{\delta}{\delta z} \end{bmatrix} \times \begin{bmatrix} 2\,yz + xz - x^2 \\ -2\,xz - yz + y^2 \\ y^2 - x^2 + z^2 \end{bmatrix} = \mathbf{V?}$$

```
> W:=<-x^2+x*z+2*y*z,y^2-2*x*z-y*z,-x^2 + y^2 + z^2>
```

$$W := \begin{bmatrix} -x^2 + x\,z + 2\,y\,z \\ -2\,x\,z + y^2 - y\,z \\ -x^2 + y^2 + z^2 \end{bmatrix} \tag{49}$$

```
> V?:=<diff(W[3],y)-diff(W[2],z),-diff(W[3],x)+diff(W[1],z),diff(W[2],x)-diff(W[1],y)>
```

(50)

$$V? := \begin{bmatrix} 2x+3y \\ 2y+3x \\ -4z \end{bmatrix} \tag{50}$$

Form here it is seen that **W** is a vector potential for **V**

D)

```
> restart:with(LinearAlgebra):with(plots):
```

Compute the flux of **V** through F after choosing orientations of F and ∂F that fulfill the right-hand rule.

```
> V:=unapply(<2*x+3*y,2*y+3*x,-4*z>,x,y,z):
  'V(x,y,z)'=V(x,y,z)
```

$$V(x,y,z) = \begin{bmatrix} 2x+3y \\ 2y+3x \\ -4z \end{bmatrix} \tag{51}$$

```
> s:=v*cos(u),v*sin(u),4-(v*cos(u))^2-(v*sin(u))^2
```

$$s := v\cos(u), v\sin(u), 4 - v^2\cos(u)^2 - v^2\sin(u)^2 \tag{52}$$

With: $u \in [0, 2\pi]$ and $v \in [0, 2]$

```
> base:=v*cos(u),v*sin(u),0
```

$$base := v\cos(u), v\sin(u), 0 \tag{53}$$

```
> n:=simplify(CrossProduct(diff(<s>,v),diff(<s>,u)))
```

$$n := \begin{bmatrix} 2v^2\cos(u) \\ 2v^2\sin(u) \\ v \end{bmatrix} \tag{54}$$

```
> n_base:=simplify(CrossProduct(diff(<base>,u),diff(<base>,v)))
```

$$n\_base := \begin{bmatrix} 0 \\ 0 \\ -v \end{bmatrix} \tag{55}$$

```
> Flux_shell:=int(V(s).n,u=0..2*Pi,v=0..2)
```

$$Flux\_shell := 0 \tag{56}$$

```
> Flux_base:=int(V(base).n_base,u=0..2*Pi,v=0..2)
```

$$Flux\_base := 0 \tag{57}$$

```
> Flux:=Flux_base+Flux_shell
```

$$Flux := 0 \tag{58}$$

```
> surface:=plot3d(<s>,u=0..2*Pi,v=0..2,style=patchnogrid,color=
  "Nautical GrayViolet",transparency=0.5):
> Base:=plot3d(<base>,u=0..2*Pi,v=0..2,style=patchnogrid,color=
  green):
> vector_field:=fieldplot3d(V(x,y,z),x=-2..2,y=-2..2,z=-1..5,
  arrows=`3-D`,grid=[7,7,7],color=red):
> display(vector_field,surface,Base,scaling=constrained,labels=[x,
  y,z],view=[-2..2,-2..2,-1..4])
```



Thus the flux throught the surface is zero

Problem 4

A)

To find the parametric representation F_r we need to combine the flow curve $r(t)$ with the parametrization $s(u, v)$ considering the initial condition $r(0) = (x_0, y_0, z_0)$.

```
> restart:with(LinearAlgebra):with(plots):
> F:=unapply(<exp(t)*((sin(u)*cos(v)+1)*cos(t)-(cos(u)+1)*sin(t)),
  (sin(u)*sin(v)+1)*exp(-t),exp(t)*((sin(u)*cos(v)+1)*sin(t)+(cos
  (u)+1)*cos(t))>,t):
  'F(t)'=F(t)
```

$$F(t) = \begin{bmatrix} e^t\,((\sin(u)\cos(v)+1)\cos(t) - (\cos(u)+1)\sin(t)) \\ (\sin(u)\sin(v)+1)\,e^{-t} \\ e^t\,((\sin(u)\cos(v)+1)\sin(t) + (\cos(u)+1)\cos(t)) \end{bmatrix} \quad (59)$$

```
> s:=<sin(u)*cos(v)+1,sin(u)*sin(v)+1,cos(u)+1>
```

$$s := \begin{bmatrix} \sin(u)\cos(v)+1 \\ \sin(u)\sin(v)+1 \\ \cos(u)+1 \end{bmatrix} \quad (60)$$

Both with: $u \in [0, \pi]$ and $v \in [0, 2\pi]$

The plot is:

```
> surface:=plot3d(s,u=0..Pi,v=0..2*Pi,style=patchnogrid,color=red):
> F1_2:=plot3d(F(1/2),u=0..Pi,v=0..2*Pi,style=patchnogrid,color=
  "Nautical GrayViolet",transparency=0.5):
> display(surface,F1_2,labels=[x,y,z],scaling=constrained)
```



B)

As was done in part A it is possible to find a parametrization for K_t combining $r(t)$ with the parametrization $r(u, v, w)$ as follows:

```
> r:=<w*sin(u)*cos(v)+c__1,w*sin(u)*sin(v)+c__2,w*cos(u)+c__3>
```

$$r := \begin{bmatrix} w \sin(u) \cos(v) + c_1 \\ w \sin(u) \sin(v) + c_2 \\ w \cos(u) + c_3 \end{bmatrix} \quad (61)$$

```
> K:=unapply(<exp(t)*((w*sin(u)*cos(v)+c__1)*cos(t)-(w*cos(u)+c__3)
  *sin(t)),(w*sin(u)*sin(v)+c__2)*exp(-t),exp(t)*((w*sin(u)*cos(v)+
  c__1)*sin(t)+(w*cos(u)+c__3)*cos(t))>,t):
  'K(t)'=K(t)
```

(62)

$$K(t) = \begin{bmatrix} e^t \left(\left(w \sin(u) \cos(v) + c_1 \right) \cos(t) - \left(w \cos(u) + c_3 \right) \sin(t) \right) \\ \left(w \sin(u) \sin(v) + c_2 \right) e^{-t} \\ e^t \left(\left(w \sin(u) \cos(v) + c_1 \right) \sin(t) + \left(w \cos(u) + c_3 \right) \cos(t) \right) \end{bmatrix} \quad \textbf{(62)}$$

The spatial surface Ωk(t) that after time t is added to the solid sphere, is thus already parameterized; We only have to put w = a in the flow line parameterizations above.

```
> Omega[K]:=unapply(<exp(t)*((a*sin(u)*cos(v)+c__1)*cos(t)-(a*cos(u)+c__3)*sin(t)),(a*sin(u)*sin(v)+c__2)*exp(-t),exp(t)*((a*sin(u)*cos(v)+c__1)*sin(t)+(a*cos(u)+c__3)*cos(t))>,t):
'Omega[K](t)'=Omega[K](t)
```

$$\Omega_K(t) = \begin{bmatrix} e^t \left(\left(a \sin(u) \cos(v) + c_1 \right) \cos(t) - \left(a \cos(u) + c_3 \right) \sin(t) \right) \\ \left(a \sin(u) \sin(v) + c_2 \right) e^{-t} \\ e^t \left(\left(a \sin(u) \cos(v) + c_1 \right) \sin(t) + \left(a \cos(u) + c_3 \right) \cos(t) \right) \end{bmatrix} \quad \textbf{(63)}$$

With: $u \in [0, \pi]$, $v \in [0, 2\pi]$ and $t \in [0, T]$

To find the volume first we need to compute the Jacobian and the do the triple integration as follows:

$$\int_K \mathrm{d}\mu = \int_0^T \int_0^{2\pi} \int_0^{\pi} Jacobian \;\mathrm{d}u\, \mathrm{d}v\, \mathrm{d}t$$

```
> Jaco:=simplify(Determinant(<diff(Omega[K](t),u)|diff(Omega[K](t),v)|diff(Omega[K](t),t)>)) assuming t>0, a>0
```

$$Jaco := -2 \left(\left(\cos(v)^2 a - a \right) \cos(u)^2 + \left(-\frac{c_1}{2} - \frac{c_3}{2} \right) \cos(u) - \cos(v)^2 a - \frac{\sin(u) \left(c_1 - c_3 \right) \cos(v)}{2} + \frac{\sin(u) \sin(v)\, c_2}{2} + \frac{a}{2} \right) \sin(u)\, a^2 e^t \quad \textbf{(64)}$$

```
> Volume:=int(Jaco,u=0..Pi,v=0..2*Pi,t=0..T) assuming T>0
```

$$Volume := \frac{4 a^3 \pi \left(-1 + e^T \right)}{3} \quad \textbf{(65)}$$

From here it is seen that the volume does not depend on the values of c_1, c_2, c_3.

The divergence can be found using theorem 28.13 in eNote 28:

$$Div(V)\left(x_0, y_0, z_0\right) = \lim_{a \to 0} \left(\frac{1}{Vol(K)} \frac{\mathrm{d}}{\mathrm{d}\,t} \left(Vol_{\pm}(t)_{|t=0} \right) \right)$$

```
> numerator:=unapply(diff(int(Jaco,u=0..Pi,v=0..2*Pi,t=0..T),T),T)
```

$$numerator := T \mapsto \frac{4 \cdot a^3 \cdot \pi \cdot e^T}{3} \quad \textbf{(66)}$$

```
> denominator:=4/3*Pi*a^3
```

$$denominator := \frac{4 a^3 \pi}{3} \quad \textbf{(67)}$$

```
> limit(numerator(0)/denominator,a=0)
```

$$1 \quad \textbf{(68)}$$